Symbolic Summation

Victor Adamchik

Carnegie Mellon University

Zeilberger's algorithm

"Programming is even more fun than proving... " -

Doron Zeilberger

Intuition

We certainly cannot expect that a first order difference operator is only what we need for proving identities. In general, we consider a higher order linear operator with polynomial coefficients $L(\Delta)$ that vanishes on a particular function f:

$$L(\Delta_n)\, f = 0$$

where

$$\Delta_n\, f = f(n+1) - f(n)$$

In a context of hypergeometric summation this means that we want to find such $L(\Delta_n)$ that will vanish on a particular definite sum

$$L(\Delta_n) \sum_k F(n, k) = \sum_k L(\Delta_n)\, F(n, k) = 0$$

where

$$L(\Delta_n)\, F(n, k) \equiv \sum_{j=0}^{N} d_j(n)\, F(n+j, k) \tag{1}$$

with unknown polynomial coefficients $d_j(n)$. Order N is also unknown. However, the existence of such N is assured under the same assumption as in Sister Celine's algorithm, namely $F(n, k)$ must be a *proper* hypergeometric term.

As in the WZ algorithm, we will be using Gosper's algorithm. Namely, we need find such a sequence $G(n, k)$ that

$$L(\Delta_n)\,F(n, k) = G(n, k+1) - G(n, k)$$

If F is a hypergeometric term in n and k with a finite support, then

$$\sum_{k=-\infty}^{\infty} L(\Delta_n)\,F(n, k) = \sum_{k=-\infty}^{\infty} \left[G(n, k+1) - G(n, k)\right] = 0 \tag{2}$$

Let us fix N and outline computational steps of Zeilberger's algorithm. The algorithms finds a linear recurrence equation for a given sum

$$S_n = \sum_{k=-\infty}^{\infty} F(n, k)$$

Denote

$$s_k = \sum_{j=0}^{N} d_j(n)\,F(n+j, k)$$

Compute the term ratio

$$\frac{s_{k+1}}{s_k} = \frac{\sum_{j=0}^{N} d_j(n)\,F(n+j, k+1)}{\sum_{j=0}^{N} d_j(n)\,F(n+j, k)} = \frac{F(n, k+1)}{F(n, k)} \frac{\sum_{j=0}^{N} d_j(n)\,\frac{F(n+j,k+1)}{F(n,k+1)}}{\sum_{j=0}^{N} d_j(n)\,\frac{F(n+j,k)}{F(n,k)}}$$

which is definitely a rational function. Find polynomials p_k, q_k and r_k and write a recurrence equation

$$q_{k+1}\, f(k) - r_k\, f(k-1) = p_k \tag{3}$$

Finally, calculate the degree bound M for $f(k)$

$$f(k) = \sum_{j=0}^{M} c_j\, k^j$$

If such a polynomial exists, we are able to find a sequence $G(n, k)$ that (2) holds. Therefore, the algorithm yields a linear difference equation with a polynomial coefficients

$$\sum_{j=0}^{N} d_j(n)\,S_{n+j} = 0 \tag{4}$$

Indeed,

$$\sum_{k=-\infty}^{\infty} L(\Delta_n)\, F(n,\ k) = \sum_{k=-\infty}^{\infty} \sum_{j=0}^{N} d_j(n)\, F(n+j,\ k) = \sum_{j=0}^{N} d_j(n) \sum_{k=-\infty}^{\infty} F(n+j,\ k) = \sum_{j=0}^{N} d_j(n)\, S_{n+j}$$

We will address the problem of solving (4) in the next lecture.

Note, when we solve equation (3) we end up with a system of linear equations wrt $M+N$ unknown variables d_j and c_j. If the system has only a trivial solution, then there is no recurrence (1) of the assumed order N. In this case, we will need to increase the order.

■ Example 1

Find a closed form for

$$S(n) = \sum_{k=0}^{n} (-1)^k \binom{n}{k}^3$$

We start with a recurrence of order $N = 1$

$$d_0(n)\, F(n,\ k) + d_1(n)\, F(n+1,\ k) = G(n,\ k+1) - G(n,\ k) \qquad (5)$$

where

$$F(n,\ k) = (-1)^k \binom{n}{k}^3$$

In order to find sequence $G(n,\ k)$ let us denote a LHS of (5) by

$$s_k = d_0(n)\, F(n,\ k) + d_1(n)\, F(n+1,\ k)$$

and apply Gosper's algorithm. Compute the ratio

$$\frac{s_{k+1}}{s_k} = \frac{(k-n-1)^3}{(k+1)^3}\ \frac{(k-n)^3\, d_0 - (n+1)^3\, d_1}{(k-n-1)^3\, d_0 - (n+1)^3\, d_1}$$

Here is a set of polynomials

$$p_k = (k-n-1)^3\, d_0 - (n+1)^3\, d_1$$

$$q_{k+1} = (k-n-1)^3$$

$$r_k = k^3$$

A recurrence $(q_{k+1}\ f_k - r_k\ f_{k-1} = p_k)$

$$(k-n-1)^3\ f(k) - k^3\ f(k-1) = (k-n-1)^3\, d_0 - (n+1)^3\, d_1$$

If it has a polynomial solution it must be of order 1. Substituting

$$f(k) = c_0 + c_1 k$$

into the above equation, we obtain the following system

$$\begin{gathered} c_0 - d_0 - d_1 = 0, \\ 3 c_0 - c_1(n+1) - 3 d_0 = 0, \\ c_0 - c_1(n+1) - d_0 = 0, \\ 3 c_1(n+1) + d_0 = 0 \end{gathered}$$

that has only a trivial solution,

We have to increase the order of (5) and repeat all the above steps. Let $N = 2$.

$$d_0(n) F(n, k) + d_1(n) F(n+1, k) + d_2(n) F(n+2, k) = G(n, k+1) - G(n, k) \tag{6}$$

The ratio

$$\frac{s_{k+1}}{s_k} = \frac{(k-n-2)^3}{(k+1)^3} \left((k-n-1)^3 \left((k-n)^3 d_0 - (n+1)^3 d_1\right) + (n+1)^3 (n+2)^3 d_2\right) \Big/$$
$$(k-n-2)^3 \left((k-n-1)^3 d_0 - (n+1)^3 d_1\right) + (n+1)^3 (n+2)^3 d_2$$

A set of polynomials

$$p_k = (k-n-2)^3 \left((k-n-1)^3 d_0 - (n+1)^3 d_1\right) + (n+1)^3 (n+2)^3 d_2$$

$$q_{k+1} = (k-n-2)^3$$

$$r_k = k^3$$

A recurrence

$$(k-n-2)^3 f(k) - k^3 f(k-1) = p_k$$

has a polynomial solution of order 4. Here is a system of equations wrt unknowns c_k and d_k

$$\begin{aligned} &(3n+2) c_4 + d_0 = 0 \\ &(n+1)^3 (d_0 + d_1 + d_2) + c_0 = 0 \\ &(2n+3) d_0 - (n+1) c_3 + \left(n^2 + 4n + 2\right) c_4 == 0 \\ &3 d_1 (n+1)^3 + 3 (2n+3) d_0 (n+1)^2 + 3 c_0 - (n+2) c_1 = 0 \\ &3 d_1 (n+1)^3 + 3 \left(5 n^2 + 15 n + 11\right) (n+1) d_0 + 3 c_0 - 3 (n+2) c_1 + (n+2)^2 c_2 = 0, \\ &d_1 (n+1)^3 + (2n+3) \left(10 n^2 + 30 n + 21\right) d_0 - \\ &\quad (3n+5) c_1 + \left(3 n^2 + 12 n + 11\right) c_2 - \left(n^2 + 5n + 7\right) (n+1) c_3 - c_4 = 0 \\ &3 \left(5 n^2 + 15 n + 11\right) d_0 + (3n+4) c_2 - 3 (n+1)(n+3) c_3 + \left(n^3 + 6 n^2 + 12 n + 4\right) c_4 = 0 \end{aligned}$$

Its solution is

$$\begin{aligned} c_0 &= 1 \\ c_1 &= -3\,(n+1)^2\left(26\,n^2+51\,n+22\right) \\ c_2 &= 3\,(n+1)\left(29\,n^2+58\,n+26\right) \\ c_3 &= -3\left(15\,n^2+32\,n+16\right) \\ c_4 &= 3\,(3\,n+4) \\ d_0 &= -\,3\,(3\,n+2)\,(3\,n+4) \\ d_1 &= 0 \\ d_2 &= -(n+2)^2 \end{aligned}$$

We won't compute $z_k = \dfrac{r_k}{p_k}\, s_k\, f_{k-1}$ but rather derive a recurrence equation for $S(n)$. Since

$$d_0(n)\,F(n,\,k)+d_1(n)\,F(n+1,\,k)+d_2(n)\,F(n+2,\,k) = G(n,\,k+1)\,-\,G(n,\,k)$$

we obtain

$$(n+2)^2\,S(n+2)\,+\,3\,(3\,n+2)\,(3\,n+4)\,S(n) = 0$$

or

$$S(n+2)\,=-\,\frac{3\,(3\,n+2)\,(3\,n+4)}{(n+2)^2}\,S(n)$$

Initial conditions

$$\begin{aligned} S(0) &= 1 \\ S(1) &= 0 \end{aligned}$$

The above equation can be easily solve by iteration

$$S(n) = \cos\left(\frac{n\,\pi}{2}\right)\frac{\left(\frac{3\,n}{2}\right)!}{\left(\frac{n}{2}\right)!^3}$$

■ *Mathematica* session

Find a closed form for

$$S(n) = \sum_{k=0}^{n} (-1)^k \binom{n}{k}^3 = \sum_{k=-\infty}^{\infty} (-1)^k \binom{n}{k}^3$$

```
F[n_, k_] := (-1)^k Binomial[n, k]^3
```

```
s[k_] := d0 F[n, k] + d1 F[n + 1, k] + d2 F[n + 2, k]
```

```
s[k + 1]/s[k] // FunctionExpand // Simplify
```

```
((-2 + k - n)^3
   (-d1 (-1 + k - n)^3 (1 + n)^3 + d2 (2 + 3 n + n^2)^3 + d0 (-k + k^2 + n - 2 k n + n^2)^3)) /
 ((1 + k)^3 ((1 + n)^3 (-d1 (-2 + k - n)^3 + d2 (2 + n)^3) +
     d0 (2 + k^2 + 3 n + n^2 - k (3 + 2 n))^3))
```

```
p[k_] :=
 ((1 + n)^3 (-d1 (-2 + k - n)^3 + d2 (2 + n)^3) + d0 (2 + k^2 + 3 n + n^2 - k (3 + 2 n))^3)
q[k_] := (-3 + k - n)^3
r[k_] := k^3
```

```
q[k + 1] f[k] - r[k] f[k - 1] - p[k]
```

```
- (1 + n)^3 (-d1 (-2 + k - n)^3 + d2 (2 + n)^3) -
 d0 (2 + k^2 + 3 n + n^2 - k (3 + 2 n))^3 - k^3 f[-1 + k] + (-2 + k - n)^3 f[k]
```

This has a polynomial solution of order 4

```
Exponent[p[k], k] - Exponent[q[k + 1] + r[k], k] + 1
```

```
4
```

```
q[k + 1] f[k] - r[k] f[k - 1] - p[k] /. f[k_] → c4 k^4 + c3 k^3 + c2 k^2 + c1 k + c0;
```

```
Collect[%, k];
CoefficientList[%, k];
Map[Equal[#, 0] &, %];
```

```
Solve[%, {c0, c1, c2, c3, c4, d0, d1, d2}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$\left\{c1 \to -\left(d0\left(-22-95n-150n^2-103n^3-26n^4\right)\right)\Big/\left(8+18n+9n^2\right),\right.$$
$$c0 \to -\left(2\,d0\left(14+71n+143n^2+143n^3+71n^4+14n^5\right)\right)\Big/\left(3\left(8+18n+9n^2\right)\right),$$
$$c2 \to -\left(d0\left(26+84n+87n^2+29n^3\right)\right)\Big/\left((2+3n)(4+3n)\right),$$
$$\left. d2 \to -\frac{d0\left(-4-4n-n^2\right)}{3(2+3n)(4+3n)},\ c3 \to -\frac{d0(-4-5n)}{2+3n},\ d1 \to 0,\ c4 \to -\frac{d0}{2+3n}\right\}$$

```
d0 S[n] + d1 S[n + 1] + d2 S[n + 2] /. %
```

$$d0\,S[n] - \left(d0\left(-4-4n-n^2\right)S[2+n]\right)\Big/\left(3(2+3n)(4+3n)\right)$$

```
RSolve[{% == 0, S[0] == 1, S[1] == 0}, S[n], n]
```

$$\left\{\left\{S[n] \to \left(3\,i^n\left(1+(-1)^n\right)\mathrm{Gamma}\left[\frac{3n}{2}\right]\right)\Big/\left(2\,\mathrm{Gamma}\left[1+\frac{n}{2}\right]^2\mathrm{Gamma}\left[\frac{n}{2}\right]\right)\right\}\right\}$$

$$S(n) = \cos\left(\frac{n\pi}{2}\right)\frac{\left(\frac{3n}{2}\right)!}{\left(\frac{n}{2}\right)!^3}$$

■ Example 2 - *Mathematica* session

Find a closed form for

$$S(n) = \sum_{k=0}^{n} (-1)^k \binom{n}{k} \binom{3k}{n}$$

```
F[n_, k_] := (-1)^k Binomial[n, k] Binomial[3 k, n]
s[k_] := d0 F[n, k] + d1 F[n + 1, k] + d2 F[n + 2, k]
s[k + 1]/s[k] // FunctionExpand // Simplify
```

```
p[k_] :=
  (-(3 k - n) (d1 (-2 + k - n) + d2 (1 - 3 k + n)) + d0 (2 + k^2 + 3 n + n^2 - k (3 + 2 n)))
q[k_] := 3 (2 + 9 (k - 1) + 9 (k - 1)^2) (-2 + (k - 1) - n)
r[k_] := (1 + 3 (k - 1) - n) (2 + 3 (k - 1) - n) (3 + 3 (k - 1) - n)
```

```
(p[k + 1] q[k + 1])/(p[k] r[k + 1]) - s[k + 1]/s[k] // FullSimplify
```

```
0
```

```
q[k + 1] f[k] - r[k] f[k - 1] - p[k] /. f[k_] → c
Collect[%, k];
CoefficientList[%, k];
Map[Equal[#, 0] &, %]
```

$$\{-12\,c - 2\,d0 - 4\,c\,n - 3\,d0\,n + 2\,d1\,n - d2\,n + 3\,c\,n^2 - d0\,n^2 + d1\,n^2 - d2\,n^2 + c\,n^3 == 0,\ -54\,c - 6\,d1 + 3\,d2 - 45\,c\,n - 4\,d1\,n + 6\,d2\,n - 9\,c\,n^2 + d0\,(3 + 2\,n) == 0,\ -d0 + 3\,d1 - 9\,d2 == 0\}$$

```
sols = Solve[%, {c, d0, d1, d2}] // Simplify
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$\left\{\left\{c \to -\frac{d0\,(3 + 2\,n)}{9\,(1 + n)\,(2 + n)},\ d1 \to \frac{d0\,(7 + 5\,n)}{3\,(1 + n)},\ d2 \to \frac{2\,d0\,(3 + 2\,n)}{9\,(1 + n)}\right\}\right\}$$

```
Clear[F];
d2 F[n + 2, k] + d1 F[n + 1, k] + d0 F[n, k] /. sols[[1]]
```

$$d0\,F[n,k] + \frac{d0\,(7+5n)\,F[1+n,k]}{3\,(1+n)} + \frac{2\,d0\,(3+2n)\,F[2+n,k]}{9\,(1+n)}$$

$$3\,(5\,n+7)\,F(n+1,\,k)+2\,(2\,n+3)\,F(n+2,\,k)+9\,(n+1)\,F(n,\,k)=0$$

```
RSolve[{9 (1 + n) S[n] + 3 (7 + 5 n) S[n + 1] + 2 (3 + 2 n) S[n + 2] == 0,
  S[0] == 1, S[1] == -3}, S[n], n]
```

$$\{\{S[n] \to (-3)^n\}\}$$

Thus

$$S(n) = \sum_{k=0}^{n} (-1)^k \binom{n}{k} \binom{3k}{n} = (-3)^n$$

■ Proving Identities

Prove

$$\sum_{k=0}^{n} \binom{n}{k}^3 = \sum_{k=0}^{n} \binom{n}{k}^2 \binom{2k}{n}$$

References

[1] M. Petkovsek, H. Wilf, D. Zeilberger, *A = B*, Algorithms and Computations in Mathematics, AK Peters, 1996.

[2] R. L. Graham, D. E. Knuth, O. Patashnik, *Concrete Mathematics*, 2nd Ed., Addison-Wesley, 1994,

[3] W. Koepf, *Hypergeometric Summation. An Algorithmic Approach to Summation and Special Function Identities*, Vieweg, Braunschweig/Wiesbaden, 1998.